
Урок 11. Простые операции с массивами. 

В этом уроке для примеров используется массив из предыдущего урока: 
var 
  a: array [1..5] of integer := (7, 8, 3, 4, 5); 

1. Перебор всех элементов по очереди 
Используется цикл for и функции Low и High. 

  { Вывод элементов массива в прямом порядке } 
  for var i := Low(a) to High(a) do 
    writeln(a[i]); 
 
  { Вывод элементов массива в обратном порядке } 
  for var i := High(a) downto Low(a) do 
    write(a[i],' '); 
  writeln; 
 
  { Вычисление суммы элементов массива } 
  var s : integer := 0; 
  for var i := Low(a) to High(a) do 
    s := s + a[i]; 
  writeln('Сумма = ', s); 
 
  { Вычисление произведения элементов массива } 
  var m : integer := 1; 
  for var i := Low(a) to High(a) do 
    m := m * a[i]; 
  writeln('Произведение = ', m); 
 
  { Обнуление всех элементов массива } 
  for var i := Low(a) to High(a) do 
    a[i] := 0; 

2. Поиск элемента в массиве 
По очереди сравниваем элементы массива с искомым значением. Если равно, то элемент найден 
и можно завершить поиск (выйти из цикла).  

  { Поиск элемента в массиве } 
  var elem : integer := 4;    // Значение искомого элемента 
  var index : integer := 0;   // Индекс искомого элемента 
   
  for var i := Low(a) to High(a) do 
    if elem = a[i] then 
    begin 
      index := i; 
      break // выход из цикла 
    end; 
 
  if index <> 0 then 
    writeln('Элемент найден, индекс = ', index) 
  else 
    writeln('Элемент не найден'); 



3. Определение минимального (максимального) элемента и его индекса 
Предполагаем, что первый элемент и есть искомый минимум (максимум). Затем перебираем все 
элементы массива, начиная со второго. Если очередной элемент меньше минимального (больше 
максимального), то сохраняем его на место искомого минимума (максимума). 

  { Поиск минимального элемента} 
  var min_elem : integer := a[1]; // Минимальный элемент 
  var index : integer := 1;       // Индекс минимального элемента 
   
  for var i := 2 to 5 do 
    if a[i] < min_elem then 
    begin 
      min_elem := a[i]; 
      index := i 
    end; 
 
  writeln('Минимум a[', index, '] = ', min_elem); 
 
 
  { Поиск максимального элемента} 
  var max_elem : integer := a[1]; // Максимальный элемент 
  var index : integer := 1;       // Индекс максимального элемента 
   
  for var i := 2 to 5 do 
    if a[i] > max_elem then 
    begin 
      max_elem := a[i]; 
      index := i 
    end; 
 
  writeln('Максимум a[', index, '] = ', max_elem); 

